VERSITA

Central European Journal of Chemistry
DOI: 10.2478/s11532-007-0029-5
Research article
CEJC 5(3) 2007 868–879

Simple method for calculating time dependence of individual radionuclide activities in decay series

Jiří Landa*, Štefan Palágyi†

Waste Disposal Department plc,
Nuclear Research Institute Řež
250 68 Husinec-Řež 130, Czech Republic

Received 30 November 2006; accepted 2 May 2007

Abstract: A rapid method for calculating the time dependence of activities of individual radionuclides in genetically coupled decay series has been proposed. The method is based on the mathematical procedure, in which the matrix method is used for calculating a set of decay equations given in the vector form. The developed method is computerized and uses the modern Scilab software. This simple method eliminates certain drawbacks of older methods used previously for this purpose and is applicable to even solve calculations which are not easily treatable with the older methods. Some practical examples of such calculations are presented. Moreover, the new method is universal and it also enables a more general approach to the problem of the calculation of decay series in nuclear chemistry.

Keywords: radioactive, activity, activities, series, decay series, decay, decay chain, scilab, solve, vector, vector solution, matrix, calculation, natural, uranium, thorium

1 Introduction

For the calculation of activities of individual radionuclides in respective decay series, the analytical solution can be found in literature [1–9], where activities of single members of the series are computed by the method of variation of constants. As an example, the calculation for a number of radioactive atoms for the third member of a decay family $N_3(t)$ can be demonstrated:

$$N_3(t) = N_1^0 \lambda_1 \lambda_2 \left(\frac{e^{-\lambda_1 t}}{(\lambda_2 - \lambda_1)(\lambda_3 - \lambda_1)} + \frac{e^{-\lambda_2 t}}{(\lambda_1 - \lambda_2)(\lambda_3 - \lambda_2)} + \frac{e^{-\lambda_3 t}}{(\lambda_1 - \lambda_3)(\lambda_2 - \lambda_3)} \right) \quad (1)$$

* E-mail: jiri.landa@nri.cz
† E-mail: stefan.palagyi@nri.cz

where N_1^0 is the number of atoms of the 1st member of series in time $t = 0$, and λ_i = decay constant of the i^{th} member of the family ($i = 1, 2, 3$). This solution is only for $N_1^0 > 0$, initial activities other members must be zero ($N_2^0 = N_3^0 = 0$).

This solution has two disadvantages, it is supposed that any two decay constants are not equal (in the case of equality of two or more constants the solution has another form) and more importantly, it is very labor intensive. In the formula given in Eq. 1, the third member of the family is computed from the initial number of atoms of the first family member. In the case that N_2^0 and/or N_3^0 are nonzero members, $N_3(t)$ must be calculated separately from N_2^0 and/or N_3^0. The resulting $N_3(t)$ is produced by the sum of $N_3(t)$ from N_1^0, N_2^0 and/or N_3^0.

Using Eq. 1 it becomes very time consuming for such a calculation for when $n = 3$ and for n greater than 3 the procedure becomes increasingly difficult.

Recently, several authors addressed the simulation and visualization of uranium and thorium decay series for the purposes of teaching, environmental and geological exploration of a decay series [7, 8]. They obtained various graphical presentations either in the form of solid curve and column diagrams or animation [9]. It was also unambiguously confirmed that the use of numerical methods in the activity calculation of decay series systems with more than three nuclides is necessary. These methods still suffer from certain restrictions on the values of decay constants of the nuclides considered in the decay series as given above.

The simple calculation method for the activity of the members for the decay series presented in this paper is independent of the initial activities of the individual radionuclides or for the equality or non-equality of the single decay constants of interest. This method is also suitable for simultaneous calculation of branching and mutually independent decay series. We assume a nonzero activity for one or more radionuclides for a single series in the closed system. From the point of view of the calculation, it is not important whether we introduce the total activity, volume activity or specific activity.

2 Method

Eq. 2 schematically demonstrates the radioactive decay series, where the nuclide X_1 decays to the nuclide X_2, nuclide X_2 decays to X_3 etc. The λ_i denotes the individual decay constants:

$$X_1 \xrightarrow{\lambda_1} X_2 \xrightarrow{\lambda_2} X_3 \xrightarrow{\lambda_3} \tag{2}$$

Using Eq. 2 we can write down a set of equations:

$$\begin{aligned} \tfrac{dN_1}{dt} &= -\lambda_1 N_1 \\ \tfrac{dN_2}{dt} &= \lambda_1 N_1 - \lambda_2 N_2 \\ &\ldots \\ \tfrac{dN_n}{dt} &= \lambda_{n-1} N_{n-1} - \lambda_n N_n \end{aligned} \tag{3}$$

This set can be solved, for example, by the method of variation of constants.

However, the set for Eq. [3] can also be rewritten into the vector form:

$$\dot{\overrightarrow{N}} = \Lambda \overrightarrow{N} \tag{4}$$

The solution of Eq. 4 is then expressed as [10]:

$$\overrightarrow{N} = e^{t\Lambda} \overrightarrow{N_0} \tag{5}$$

where $\overrightarrow{N}$ is the vector of the number of atoms, $\dot{\overrightarrow{N}}$ is its derivation in time, and $\overrightarrow{N_0}$ is the vector of the number of atoms in $t = 0$. The matrix of Λ is:

$$\Lambda = \begin{pmatrix} -\lambda_1 & 0 & 0 & 0 & 0 & 0 \\ \lambda_1 & -\lambda_2 & 0 & 0 & 0 & 0 \\ 0 & \lambda_2 & -\lambda_3 & 0 & 0 & 0 \\ 0 & 0 & ... & ... & 0 & 0 \\ 0 & 0 & 0 & \lambda_{n-2} & -\lambda_{n-1} & 0 \\ 0 & 0 & 0 & 0 & \lambda_{n-1} & -\lambda_n \end{pmatrix} \tag{6}$$

For i^{th} radionuclide with decay by α and β we can write down a set of equations:

$$\begin{aligned} &... \\ &\tfrac{dN_i}{dt} = \lambda_{i-1} N_{i-1} - \lambda_i N_i \\ &\tfrac{dN_{i+1}}{dt} = p_{\beta i} \lambda_i N_i - \lambda_{i+1} N_{i+1} \\ &\tfrac{dN_{i+2}}{dt} = p_{\alpha i} \lambda_i N_i - \lambda_{i+2} N_{i+2} \\ &\tfrac{dN_{i+3}}{dt} = \lambda_{i+1} N_{i+1} + \lambda_{i+2} N_{i+2} - \lambda_{i+3} N_{i+3} \\ &... \end{aligned} \tag{7}$$

and relevant rows of the matrix Λ have the form:

$$\Lambda = \begin{pmatrix} ... & ... & 0 & 0 & 0 & 0 & 0 \\ 0 & \lambda_{i-1} & -\lambda_i & 0 & 0 & 0 & 0 \\ 0 & 0 & p_{\beta i}\lambda_i & -\lambda_{i+1} & 0 & 0 & 0 \\ 0 & 0 & p_{\alpha i}\lambda_i & 0 & -\lambda_{i+2} & 0 & 0 \\ 0 & 0 & 0 & \lambda_{i+1} & \lambda_{i+2} & -\lambda_{i+3} & 0 \\ 0 & 0 & 0 & 0 & 0 & ... & ... \end{pmatrix} \tag{8}$$

where $p_{\alpha i}$ and $p_{\beta i}$ are the probabilities of the α or β decay i^{th} term. An example of Λ matrix for independent series is:

$$\Lambda = \begin{pmatrix} -\lambda_{x1} & 0 & 0 & 0 & 0 & 0 & 0 \\ \lambda_{x1} & -\lambda_{x2} & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & -\lambda_{y1} & 0 & 0 & 0 & 0 \\ 0 & 0 & \lambda_{y1} & -\lambda_{y2} & 0 & 0 & 0 \\ 0 & 0 & 0 & \lambda_{y2} & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & -\lambda_{z1} & 0 \\ 0 & 0 & 0 & 0 & 0 & \lambda_{z1} & -\lambda_{z2} \end{pmatrix} \tag{9}$$

where the 'x' series has two radionuclides, the 'y' series has two radionuclides and a third stable nuclide and the 'z' series has two radionuclides.

The exponential function of the matrix is defined as [10]:

$$e^M = \sum_{k=1}^{\infty} \frac{M^k}{k!} \tag{10}$$

where $M \in (R)^{nxn}$. For the calculation of this difficult formula, we use a mathematical software which has a command for an exponential function of matrix. For each calculating time step we let $M = t\Lambda$.

3 Results and discussion

There are four natural decay series which are composed of the radionuclides heavier than ^{206}Pb [11]. According to the divisibility of the mass number of both the initial and final nuclides by 4, the individual series are denoted as 4n+0 (from ^{232}Th to ^{208}Pb), 4n+1 (from ^{237}Np to ^{209}Bi), 4n+2 (from ^{238}U to ^{206}Pb) and 4n+3 (from ^{235}U to ^{207}Pb). Nuclides from the 4n+1 series do not occur in nature due to their short half-lives in comparison with the age of the Earth, but they can be produced artificially [4]. The series 4n has 11 radionuclides, the series 4n+2 has 15 radionuclides, and the series 4n+3 has 13 radionuclides.

The time dependence of the activities of nuclides in a single series can be solved with the help of Eq. 5. It is necessary to take into consideration scale differences in the half-life, e.g., in 4n+0 the longest half-life has ^{232}Th ($1.4 \cdot 10^{10}$ years) and the shortest ^{212}Po ($1.71 \cdot 10^{-8}$ s), which is as many as 25 orders of magnitude. In the case that the computed task contains radionuclides of several series, it is better to calculate single series separately, even though Eq. 5 enables them to be calculated simultaneously.

Scilab software was used for the calculation (it also exists in the version for Win32 [12]). This software is designed for mathematical calculations, including the exponential of matrices.

For the calculation of activities of decay series, it is advisable to omit radionuclides with the shortest half-lives but to take into consideration their equilibrium activities with the mother radionuclides, e.g. ^{216}Po, ^{212}Po in 4n+0 series, ^{217}At, ^{213}Po in 4n+1 series, ^{214}Po in 4n+2 series and ^{215}Po, ^{211}Po in 4n+3 series.

For facilitating the individual calculations some functions have been created in the Scilab software. These functions are listed in the Annex (List of file "fdecay.sci"). The first example is shown for the decay series, which has two radionuclides with a given half-life and nonzero initial activities for each radionuclide. A list of the batch file for the calculation is given in the Annex (List of file "decay_st1.sci"). The result of this calculation for two unnamed radionuclides, as an output from Scilab software, is presented in Fig. 1. The half-life of both the first (mother) and the second (daughter) radionuclide were the same 1 time unit. This calculation method is equally valid for any coupled pair of radionuclides with zero or nonzero initial activity of the second (daughter) radionuclide with different half-lives.



Fig. 1 Time dependence of the activity of two genetically coupled radionuclides calculated by the elaborated method.

The second example is the calculation of three radionuclides, ^{238}U, ^{234}U and ^{226}Ra from the decay series 4n+2, present in a sample of ground water. The initial activity of ^{238}U is 5 Bq·l^{-1}. The activity ratio of ^{234}U/^{238}U is chosen as 10, i.e., the activity of ^{234}U is 50 Bq·l^{-1}, and the initial activity of ^{226}Ra is 20 Bq·l^{-1}. The other radionuclides from this series are disregarded. A list of the batch file for the calculation is listed in the Annex (List of file "decay_s2.sci") and the result of the calculation as an output from Scilab software, is given in Fig. 2.



Fig. 2 Time dependence of the activity of three genetically coupled radionuclides from uranium decay series 4n+2 calculated by the elaborated method.



Fig. 3 Time dependence of the activity of six genetically coupled radionuclides from uranium decay series 4n+3 calculated by the elaborated method.

The third example is for six radionuclides from uranium decay series 4n+3, where the third radionuclide, ^{227}Ac has two types of decay (α and β). A list of the batch file for the calculation is given in the Annex (List of file "decay_s3.sci"). For greater clarity we left out ^{231}Th. The result for the calculation presented in Fig. 3 shows that the activity was practically unchanged for ^{235}U in the time span under consideration, the activities of ^{231}Pa, ^{227}Ac and ^{223}Ra gradually attain the activity of ^{235}U, except ^{227}Th and ^{223}Fr. The sum of the activities of these last two radionuclides is equal to the activity of ^{235}U at equilibrium. The difference in the activities of ^{227}Th and ^{223}Fr is due to the greater differences of α and β decay probabilities for their ^{227}Ac mother nuclide.

4 Conclusion

A computerized numerical method has been developed. This method enables the simultaneous calculation of the activity of any member for all of the uranium decay series at any time or any time interval for various initial conditions. It has some advantages when the number of nuclides in a decay series is higher than 2. The method is universal and easy, and it makes it possible to calculate activities in cases where the initial activities of the daughter nuclides have nonzero values. It is applicable also in a hypothetic case where, in any decay series, two or more decay constants are equal. The used Scilab software allows the presentation of the calculated activity values in a direct graphical or tabulated form, as desired. The proposed method can be used advantageously in environmental or geological studies where the natural radionuclides, especially U or Th decay series are used.

Acknowledgement

This research was supported by the Ministry of Trade and Industry of the Czech Republic under contract No. 1H-PK/25 and by Czech Science Foundation under contract No. 104/06/1583.

Annex

Source code listing of the batch file "fdecay.sci":

```
// functions, version 2.4 25.1.2007 jl
//conversion of vector of half times to vector of decay constants
//input parameters:  T - vector of half times
function L=T2L(T)
L=[];
for i=1:length(T),
    if T(i)>0 then L(i)=log(2)/T(i);
    else L(i)=0;
    end;
```

```
    end;
endfunction;

//generate of decay matrix from decay vector
//input parameters:  L - vector of decay constants
function LL=L2LL(L)
LL=[];
for i=1:length(L)
   for j=1:length(L)
       LL(i,j)=0;
       if i==j then LL(i,j)=-L(i); end;
       if i-j==1 then LL(i,j)=L(j); end;
       end;
   end;
endfunction;

//correct of decay matrix for radionuclide with both type of decay
//input parameters:  LL - decay matrix, i - index of radionuclide
//                   pb - probability of betas decay of radionuclide i
function LL=LLc(LL,i,pb)
   n=size(LL)
   if n(1)>i then LL(i+1,i)=-LL(i,i)*pb; end;
   if n(1)>=i+2 then
       LL(i+2,i)=-LL(i,i)*(1-pb);
       LL(i+2,i+1)=0;
       end;
   if n(1)>=i+3 then
       LL(i+3,i+1)=-LL(i+1,i+1);
       LL(i+3,i+2)=-LL(i+2,i+2);
       end;
endfunction;

//convert of initial activities vector to initial atom number vector
//input parameters:  a - vector of activities, L - vector of decay constants
function n=A2N(a,L)
n=[];
for i=1:length(a),
   n(i)=a(i)/L(i);
   end;
endfunction;

// solve decay series by matrix method
```

```
//input parmeters:  LL - matrix of decay constants,
//n0 - initial vector of atom nubers, t - vector of time steps
function nn=decay(LL,n0,t)
nn=[];
for i=1:length(t),
   n=expm(t(i)*LL)*n0;
   for j=1:length(n0),
       nn(i,j)=n(j); end; end;
endfunction;

//convert of result atom number vector(matrix) to result activity vector(matrix)
//input parameters:  n - vector of numbers of atoms,
//L - vector of decay constants
function a=N2A(n,L)
a=[];
[ni,nj]=size(n);
for i=1:ni,
   for j=1:nj,
       a(i,j)=n(i,j)*L(j);
   end;
   end;
endfunction;
```

End of the list of file ”fdecay.sci”:

Source code listing of the batch file ”decay_st1.sci”:

```
//load of the functions
;exec("[PATH]\ fdecay.sci");

//declares
T=[ //declare vector of half times of decay series
1; //half time of the first radionuclide
1; //half time of the second radionuclide
];

a0=[1000 400]'; //declare vector of initial activities

t=[0:.1:10]'; //declare of time vector

//conversion and calculates
L=T2L(T); //conversion of vector of half times to vector of decay constants

LL=L2LL(L); //conversion of vector of decay constant to the matrix
```

```
//conversion vector of initial activities to vector of initial atom numbers:
n0=A2N(a0,L);

nn=decay(LL,n0,t); //calculate of vectors of atom number in time steps

na=N2A(nn); //conversion vectors of atom numbers to activities


// plot of the results
xbasc() //reset of graph window
xset("default"); //default parameters of graph window
xtitle(",'time [units]','activity [Bq]') //set graph titles
//plot
plot2d(t,na,style=[-1,-2]...
leg="The first radionuclide@The second
radionuclide");
```

End of the list of file "decay_st1.sci".

Source code listing of the batch file "decay_s2.sci":

```
//load of the functions
;exec("[PATH]\fdecay.sci");

//declares
sy=365.24219*24*60*60; //sec per year

T=[ // declare vector of half times [s]
4.46e9*sy; //U
-238
2.45e5*sy; //U -234
1600*sy; //Ra-226
];


t=[0:400*sy:16000*sy]'; //declare of time vector

a0=[5 50 20]'; //declare vector of initial activities [Bq/l]


//conversion and calculates
L=T2L(T); //conversion halftime to decay constants
LL=L2LL(L); //conversion decay constant vector to the matrix
```

```
n0=A2N(a0,L); //conversion initial activities to number of atoms

nn=decay(LL,n0,t); //calculate of atom number vectors in time steps
na=N2A(nn); //conversion atom numbers to activities
// plot of the results
xbasc() //reset of graph window
xset("default"); //default parameters of graph window
xtitle(",'time [y]','activity [Bq/l]') //set graph titles
plot2d(t/sy,na,logflag='nn',style=[-1 -2 -3], leg="U238@U234@Ra226"); //plot
```

End of the list of file "decay_s2.sci".

Source code listing of the batch file "decay_s3.sci":

```
//load of the functions
;exec("[PATH]\fdecay.sci");

//declares
T=[ //declare vector of half times of 4n+3 decay series [y]
7.03e8;                 //U -235 1
3.27e4;                 //Pa-231 2
21.8;                   //Ac-227 3
18.9/365.24219;         //Th-227 4
22/60/24/365.24219;     //Fr-223 5
11.4/365.24219;         //Ra-223 6
];

ns3=3; // index of Ac-227
ps3b=0.986; // probability of beta decay of Ac-227

a0=[950 0 0 0 0 0]'; //declare vector of initial activities

t=[0:6e3:3e5]'; //declare of time vector

//conversion and calculates
L=T2L(T); //conversion of vector of half times to vector of decay constants

LL=L2LL(L); //conversion of vector of decay constant to the matrix
LL=LLc(LL,ns3,ps3b); //correct the matrix for alpha and beta decay of Ac-227

//conversion vector of initial activities to vector of initial atom numbers
n0=A2N(a0,L);
nn=decay(LL,n0,t); //calculate of vectors of atom number in time steps
```

```
na=N2A(nn); //conversion vectors of atom numbers to activities

// plot of the results
xbasc() //reset of graph window
xset("default"); //default parameters of graph window
xtitle(",'time [y]','activity [Bq]') //set graph titles
plot2d(t,na,logflag='nn',style=[-1 -2 -3 -4 -5 -6],...
leg="U235@Pa231@Ac227@Th227@Fr223@Ra223"); //plot
```

End of the list of file "decay_s3.sci".

References

[1] H. Bateman: "The solution of a system of differential equations occurring in the theory of radioactive transformations", *Proc. Cambridge Phil. Soc.*, Vol. 16, (1910), pp. 423–427.

[2] J.F. Duncan and G.B. Cook: *Isotopes in Chemistry*, Clarendon Press, Oxford, 1968.

[3] M. Wiernik: "Time-dependent relative activities in the radioactive families", *J. Radioanal. Chem.*, Vol. 31, (1976), pp. 529–568.

[4] G. Faure: *Principles of Isotope Geology*, 2nd ed., J. Wiley and Sons, New York, 1986.

[5] J. Tölgyessy and E. Bujdosó: *CRC Handbook of Radioanalytical Chemistry*, Vol I, CRC Press, Boca Raton, 1991.

[6] S. Mirzadeh and P. Walsh: "Numerical Evaluation of the Production of Radionuclides in a Nuclear Reactor - Part I", *Appl. Radiat. Isot.*, Vol. 49, (1998), pp. 379–382.

[7] M. Ivanovich and R.S. Harmon: *Uranium-series Disequilibrium, Applications to Earth, Marine, and Environmental Sciences*, 2nd ed., Clarendon Press, Oxford 1992.

[8] K. Rasilainen and J. Suksi: "A multisystem modelling approach for uranium-series dating", *Nucl. Technol.*, Vol. 120, (1997), pp. 254–260.

[9] S. Azzam and J. Suksi: "DECSERVIS: A tool for radioactive decay series visualization", *Centr. Europ. J. Chem.*, published online: July 4, 2006, DOI 10. 10.2478/s11532-006-0026-0.

[10] L. Motl and M. Zahradník: *Let us pursuing linear algebra* (in Czech), Karolinum Press, Prague, 2002.

[11] Ch. Hacker: Radiation Decay, Version 4, FreeWare, Griffith University, Brisbane, 2005.

[12] Scilab Reference Manual, On-line Documentation, Scilab Group, http://www.scilab.org/, http://www.scilab.org/download/index_download.php, http://www.gu.edu.au/school/eng/mmt/RadDec.html.